Министерство образования и науки Российской Федерации
Федеральное агентство по образованию
Дальневосточный государственный университет

Л.А. МОЛЧАНОВА

ЗАДАЧА ШТУРМА-ЛИУВИЛЛЯ

Методические указания для студентов
математических специальностей

Владивосток
Издательство Дальневосточного университета
2006

ББК 22.161
М75

Рецензенты:
А.Г. Колобов, к.ф.-м.н. (ИМКН ДВГУ);
М.А Князева, к.т.н. (ИАПУ ДВО РАН).

Молчанова Л.А.
М75 **Задача Штурма-Лиувилля.** Учебно-методическое пособие. - Владивосток: Изд-во Дальневост. ун-та, 2006. - 16с.

Методические указания разработаны для студентов Института математики и компьютерных наук ДВГУ. В них дается теоретический материал, позволяющий студентам использовать средства математического пакета Maple в своей практической деятельности при выполнении заданий по спецкурсу Пакеты прикладных программ и математических дисциплин, связанных с решением задачи Штурма - Лиувилля.

Для студентов математических специальностей.

$\mathrm{M}\frac{1702050000}{180(03)-2006}$ ББК 22.161

Задача Штурма-Лиувилля

Рассмотрим обыкновенное однородное линейное дифференциальное уравнение второго порядка

$$[p(x)X'(x)]' + [\lambda r(x) - q(x)]X(x) = 0, \quad a < x < b, \tag{1}$$

где $p(x)$, $q(x)$, $r(x)$ - вещественные функции от x. Предполагается, что $p(x)$, $p'(x)$, $q(x)$, $r(x)$ непрерывны в (a,b); $p(x)$ и $r(x)$ положительны в (a,b); λ - параметр, принимающий любые значения.

Концы интервала (a,b) могут быть как обыкновенными точками, так и особыми (сингулярными). Напомним, что если при некотором x хотя бы один из коэффициентов уравнения (2) имеет бесконечный разрыв или $p(x) = 0$, то говорят, что *коэффициенты уравнения имеют особенность* в точке x.

Граничная задача, в которой решения уравнения (2) удовлетворяют *однородным линейным граничным условиям с вещественными коэффициентами* называют *задачей Штурма-Лиувилля.* Таким образом, под задачей Штурма-Лиувилля понимается следующая задача: найти решение уравнения (2), принадлежащее классу $C^{(2)}(a,b)$ и удовлетворяющие некоторым однородным граничным условиям, заданным на концах интервала (a,b). Примером таких условий могут быть условия

$$X|_{x\to a+0} = 0, \quad X|_{x\to b-0} = 0.$$

Различают задачи двух типов - *регулярную задачу* и *сингулярную задачу.* Задача Штурма-Лиувилля называется регулярной, если интервал (a,b) конечен, концы интервал (a,b) - обыкновенные точки рассматриваемого уравнения. Задача называется сингулярной, если хотя бы одно из этих условий не выполнено. Сингулярная задача может быть с одним или двумя сингулярными концами. Характер однородных граничных условий регулярной и сингулярной задач разный.

Сформулируем типовые граничные условия. В регулярной задаче различают *граничные условия первого рода*

$$X(a) = 0, X(b) = 0;$$

граничные условия второго рода

$$X'(a) = 0, X'(b) = 0;$$

граничные условия третьего рода

$$X'(a) - h_a X(a) = 0, X'(b) + h_b X(b) = 0, \; h_a, h_b \geq 0;$$

граничные условия четвертого рода

$$X(a) = X(b),\ X'(a) = X'(b),\ p(a) = p(b).$$

Граничные условия четвертого рода называют *условиями периодичности.*

В случае сингулярной задачи различают два варианта задач в зависимости от того, один или два конца сингулярны. Пусть $x = a$ - сингулярный конец, $x = b$ - регулярный конец. Тогда на сингулярном конце ставится условие ограниченности функции

$$X|_{x \to a+0} = 0(1),$$

а на регулярном конце могут быть условия первого, второго или третьего рода.

Если оба конца сингулярны, то ставятся условия ограниченности функции

$$X|_{x \to a+0} = 0(1),\ X|_{x \to b-0} = 0(1).$$

Нас будут интересовать *нетривиальные* $X \neq 0$ решения задачи. Однако нетривиальных решений при данном произвольном λ может и не быть. Поэтому содержанием задачи Штурма-Лиувилля является не только отыскание решений (*собственных функций*) при данном λ, но и определение совокупности (спектра) *собственных значений* λ, при которых существует нетривиальные решения. Собственные функции, по определению, находятся с точностью до произвольной константы. Если накладывается условие

$$\int_a^b r(x)|X(x)|^2 dx = 1,$$

то имеют дело с нормированными собственными функциями.

Для регулярной задачи Штурма-Лиувилля с граничными условиями первого, второго, третьего или четвертого рода имеют место следующие теоремы.

Теорема 1. *Все собственные значения регулярной задачи Штурма - Лиувилля вещественны.*

Теорема 2. *Все собственные значения регулярной задачи Штурма - Лиувилля ограничены снизу.*

Некоторые свойства собственных значений регулярной задачи Штурма-Лиувилля

Для регулярной задачи в соответствии с общей теорией Коши, можно утверждать, что существует интеграл $X(x) \in C^{(2)}$ уравнения (1), удовлетворяющий условиям $X(c) = \alpha$, $X'(c) = \beta$, α, β - любые числа, $c \in (a, b)$.

Если числа α,β не зависят от параметра λ, то решение $X(x)$ при фиксированном x является *целой функцией* от λ.

Под *целой функцией* комплексного переменного λ подразумевается такая функция, которая разлагается в ряд Тейлора во всей области (радиус сходимости равен ∞), то есть $f(\lambda)$ - целая функция, если

$$f(\lambda) = \sum_{n=0}^{\infty} C_n \lambda^n = \sum_{n=0}^{\infty} \frac{f^{(n)}(0)}{n!} \lambda^n, |\lambda| < \infty.$$

В качестве точки $x = c$ возьмем граничную точку $x = a$. Тогда существуют два интеграла $\varphi(x, \lambda)$ и $\psi(x, \lambda)$, удовлетворяющие условиям

$$\varphi(x, \lambda) : X(a) = 1,\ X'(a) = 0,$$

$$\psi(x, \lambda) : X(a) = 0,\ X'(a) = 1.$$

Можно записать

$$\begin{aligned} &\varphi(a, \lambda) = 1,\ \varphi'(a, \lambda) = 0, \\ &\psi(a, \lambda) = 0, \psi'(a, \lambda) = 1. \end{aligned} \tag{2}$$

При этом интегралы $\varphi(x, \lambda)$, $\psi(x, \lambda) \in C^{(2)}(a, b)$ и являются, при фиксированном x, целыми функциями от λ. Эти функции линейно независимы и называются *фундаментальной системой решений Штурма-Лиувилля.*

Общий интеграл уравнения Штурма-Лиувилля (1) можно представить в виде

$$X(x, \lambda) = A\varphi(x, \lambda) + B\psi(x, \lambda),\ A, B = const. \tag{3}$$

Числа A, B и λ выбираются так, чтобы полученное решение отвечало граничным условиям. В случае условий первого рода

$$A\varphi(a, \lambda) + B\psi(a, \lambda) = 0,$$

$$A\varphi(b, \lambda) + B\psi(b, \lambda) = 0,$$

или

$$A = 0,\ \psi(b, \lambda) = 0.$$

В случае условий второго рода

$$A\varphi'(a, \lambda) + B\psi'(a, \lambda) = 0,$$

$$A\varphi'(b, \lambda) + B\psi'(b, \lambda) = 0,$$

или

$$B = 0,\ \varphi'(b, \lambda) = 0.$$

В случае условий третьего рода уравнение для определения λ имеет вид

$$\varphi'(b,\lambda) + h_b\varphi(b,\lambda) + h_a\psi'(b,\lambda) + h_a h_b\psi(b,\lambda) = 0.$$

Оказывается, что все полученные уравнения для определения λ имеет бесчисленное множество решений.

Спектр регулярной задачи Штурма-Лиувилля - счетное множество вещественных чисел без точек сгущения. Пусть $\lambda_1 < \lambda_2 < \cdots < \lambda_n$ - последовательность собственных значений. Тогда в случае условий первого рода будем иметь собственные функции

$$X_n(x) = B_n\psi(x,\lambda_n),\ n = 1, 2, 3, \cdots$$

В случае условий второго рода будем иметь

$$X_n(x) = A_n\varphi(x,\lambda_n),\ n = 1, 2, 3, \cdots$$

и в случае условий третьего рода

$$X_n(x) = A_n[\varphi_n(x,\lambda_n) + h_a\psi(x,\lambda_n)],\ n = 1, 2, 3, \cdots$$

Имеет место следующая теорема.

Теорема. *Система собственных функций регулярной задачи Штурма - Лиувилля ортогональна на*[a,b] *с весом* $r(x)$, то есть

$$\int_a^b r(x)X_m(x)X_n(x)dx = \begin{cases} 0, & m \neq n, \\ ||X_n||^2, & m = n. \end{cases}$$

Литература

1. Говорухин В.Н., Цибулин В.Г. Компьютер в математическом исследовании. Учебный курс. - СПб.: Питер, 2001.

2. Голоскоков Д. П. Уравнения математической физики. Решение задач в системе Maple. Учебник для вузов. -СПб.: Питер, 2004.

3 Будак Б.М., Самарский А.А., Тихонов А.Н. Сборник задач по математической физике. М.: Наука, 1972.

4. Сборник задач по математике для втузов. Специальные курсы. М.: Наука, 1984.

5. Смирнов М.М. Задачи по уравнениям математической физики. М.: Наука, 1975.

6. Тихонов А.Н., Самарский А.А. Уравнения математической физики. М.: Наука, 1984.

Задания

1 Найти собственные значения и собственные функции краевой задачи

1. $y'' + \lambda y = 0,\ y(0) = 0,\ y(l) = 0,\ x \in [0, l];$
2. $y'' + \lambda y = 0,\ y(0) = 0,\ y'(l) + hy(l) = 0,\ x \in [0, l],\ h > 0;$
3. $y'' + \lambda y = 0,\ y(l) = 0,\ -y'(0) + hy(0) = 0,\ x \in [0, l],\ h > 0;$
4. $y'' + \lambda y = 0,\ y(0) = 0,\ y'(l) = 0,\ x \in [0, l];$
5. $y'' + \lambda y = 0,\ y'(0) = 0,\ y(l) = 0,\ x \in [0, l];$
6. $y'' + \lambda y = 0,\ y'(0) = 0,\ y'(l) = 0,\ x \in [0, l];$
7. $y'' + \lambda y = 0,\ y'(0) = 0,\ y'(l) + hy(l) = 0,\ x \in [0, l],\ h > 0;$
8. $y'' + \lambda y = 0,\ y'(l) = 0,\ -y'(0) + hy(0) = 0,\ x \in [0, l],\ h > 0;$
9. $y'' + \lambda y = 0,\ -y'(0) + hy(0) = 0,\ y'(l) + hy(l) = 0,\ x \in [0, l],\ h > 0;$
10. $(xy')' + \lambda/xy = 0,\ y(a) = 0,\ y(b) = 0,\ x \in [a, b],\ a \neq 0,\ b \neq 0;$
11. $(x^2y')' + \lambda x^2 y = 0,\ y(a) = 0,\ y(b) = 0,\ x \in [a, b],\ a \neq 0,\ b \neq 0;$
12. $y'' + \lambda y = 0,\ y(-l) = 0,\ y(l) = 0,\ x \in [-l, l];$
13. $y'' + \lambda y = 0,\ y'(-l) = 0,\ y'(l) = 0,\ x \in [-l, l];$
14. $y'' + \lambda y = 0,\ -y'(-l) + hy(-l)=0,\ y'(l) + hy(l)=0,\ x \in [-l, l], h > 0;$
15. $(xy')' + \lambda/xy = 0,\ y'(a) = 0,\ y'(b) = 0,\ x \in [a, b],\ a \neq 0,\ b \neq 0;$
16. $((x + 0.2\cos\pi x)y')' + \lambda y = 0,\ y(0) - y'(0) = 0,\ y(1) = 0,\ x \in [0, 1];$
17. $((x + 0.2\cos\pi x)y')' + \lambda y = 0,\ y(0) = 0,\ y(1) + y'(1) = 0,\ x \in [0, 1];$
18. $((x + 0.2\cos\pi x)y')' + \lambda y = 0,\ y(0) - y'(0) = 0\, y'(1) = 0,\ x \in [0, 1];$
19. $((x + 0.2\cos\pi x)y')' + \lambda y = 0,\ y'(0) = 0\, y(1) + y'(1) = 0,\ x \in [0, 1];$
20. $((x + 0.2\cos\pi x)y')' + \lambda y = 0,\ y(0) = 0\, y(1) + y'(1) = 0,\ x \in [0, 1];$
21. $((x + 0.2\cos\pi x)y')' + \lambda y = 0,\ 2y(0) + y'(0) = 0\, y(1) = 0,\ x \in [0, 1];$
22. $((x + 0.2\cos\pi x)y')' + \lambda y = 0,\ y(0) - y'(0) = 0,\ y(1) = 0,\ x \in [0, 1];$
23. $((4x + \cos\pi x)y')' + \lambda y = 0,\ y(0) = 0,\ y(1) + y'(1) = 0,\ x \in [0, 1];$
24. $((4x + \cos\pi x)y')' + \lambda y = 0,\ y(0) - y'(0) = 0\, y'(1) = 0,\ x \in [0, 1];$
25. $((4x + \cos\pi x)y')' + \lambda y = 0,\ y'(0) = 0\, y(1) + y'(1) = 0,\ x \in [0, 1].$

Примеры решения типовых задач

Пример 1. Найти собственные значения и собственные функции задачи Штурма-Лиувилля:

$$(x^2y')' + \lambda x^2 y = 0,\ y(0) = O(1),\ y'(a) = 0,\ x \in [0, a].$$

Решение. Зададим уравнения при $\lambda \neq 0$ и $\lambda = 0$:

```
>restart:eq:=diff(x^2*diff(y(x),x),x)+lambda*x^2*y(x)=0;
>lambda:=0;eq0:=subs(y(x)=y0(x),eq);lambda:='lambda';
```

$$eq := 2\,x\,(\frac{d}{dx}\,\mathrm{y}(x)) + x^2\,(\frac{d^2}{dx^2}\,\mathrm{y}(x)) + \lambda\,x^2\,\mathrm{y}(x) = 0$$

$$\lambda := 0$$

$$eq0 := 2\,x\,(\frac{d}{dx}\,\mathrm{y0}(x)) + x^2\,(\frac{d^2}{dx^2}\,\mathrm{y0}(x)) = 0$$

$$\lambda := \lambda$$

Находим общее решение уравнения при $\lambda \neq 0$:

```
>dsol:=dsolve(eq,y(x));assign(dsol):
```

$$dsol := \mathrm{y}(x) = \frac{\_C1\,\sin(\sqrt{\lambda}\,x)}{x} + \frac{\_C2\,\cos(\sqrt{\lambda}\,x)}{x}$$

Проверим правильность этого решения:

```
>simplify(value(eq));
```

$$0 = 0$$

Видим, что решение будет ограничено в нуле, если принять константу при косинусе равной нулю:

```
y:=subs(_C2=0,y(x));
```

$$y := \frac{\_C1\,\sin(\sqrt{\lambda}\,x)}{x}$$

Находим производную:

```
>y1:=simplify(diff(y,x));
```

$$y1 := \frac{\_C1\,(-\sin(\sqrt{\lambda}\,x) + \cos(\sqrt{\lambda}\,x)\,\sqrt{\lambda}\,x)}{x^2}$$

Получаем уравнение для определения собственных значений:

```
>eq1:=expand(subs(x=a,y1)/_C1*a^2)=0;
>eq1:=expand(lhs(eq1)/cos(lambda^(1/2)*a))=0;
```

$$eq1 := -\sin(\sqrt{\lambda}\, a) + \cos(\sqrt{\lambda}\, a)\, \sqrt{\lambda}\, a = 0$$

$$eq1 := -\frac{\sin(\sqrt{\lambda}\, a)}{\cos(\sqrt{\lambda}\, a)} + \sqrt{\lambda}\, a = 0$$

Преобразовываем это уравнение и определяем функцию, нулями которой задаются собственные значения:

```
>eq1:=convert(eq1,tan);
>f:=subs(lambda^(1/2)*a=mu,lhs(eq1));
```

$$eq1 := -\tan(\sqrt{\lambda}\, a) + \sqrt{\lambda}\, a = 0$$

$$f := -\tan(\mu) + \mu$$

Здесь мы ввели обозначение $f(\mu) = -tg(\mu) + \mu$, причем $\mu = \sqrt{\lambda}a$.
Пусть $\mu_k, k = 1, 2, 3, \cdots$ - положительные корни уравнения $f(\mu) = 0$. Тогда собственные значения задачи определяются по формуле $\lambda_k = (\mu_k/a)^2$. Собственные функции будут $\sin(\mu_k x/a)/x$. Определяем эти функции:

```
>y:=(x,n)->sin((mu[n]/a)*x)/x;
```

$$y := (x,\, n) \to \frac{\sin(\frac{\mu_n\, x}{a})}{x}$$

Согласно общей теории, эти функции ортогональны на [0,a] с весом x^2. Проверим это:

```
>Int(x^2*y(x,n)*y(x,m),x=0..a);res:=value(%);
>simplify(res,{sin(mu[m])=mu[m]*cos(mu[m]),
>sin(mu[n])=mu[n]*cos(mu[n])});
```

$$\int_0^a \sin(\frac{\mu_n\, x}{a}) \sin(\frac{\mu_m\, x}{a})\, dx$$

$$0$$

Вычислим квадрат нормы:

```
>Int(x^2*y(x,n)^2,x=0..a);Norma:=simplify(value(%));
```

$$\int_0^a \sin(\frac{\mu_n\, x}{a})^2\, dx$$

$$Norma := \frac{1}{2}\, \frac{a\, (\cos(\mu_n) \sin(\mu_n) - \mu_n)}{\mu_n}$$

Упростим результат:

```
>Norma:=subs(sin(mu[n])=mu[n]*cos(mu[n]),Norma);
>Norma:=simplify(Norma,{cos(mu[n])^2=-sin(mu[n])^2+1});
```

$$Norma := \frac{1}{2}\,\frac{a\,(-\cos(\mu_n)^2\,\mu_n + \mu_n)}{\mu_n}$$

$$Norma := \frac{1}{2}\,a\,\sin(\mu_n)^2$$

Проверим, будет ли $\lambda = 0$ собственным значением задачи:

```
>sol0:=dsolve(eq0,y0(x));assign(sol0):simplify(value(eq0));
```

$$sol0 := y0(x) = \_C1 + \frac{\_C2}{x}$$

$$0=0$$

Видим, что это решение будет ограниченным в нуле, если $\_C2 = 0$, а тогда оно удовлетворяет второму краевому условию при любом $\_C1$. Таким образом, получаем, что $\lambda = 0$ является собственным значением задачи, которому отвечает собственная функция, равная единице.

Посмотрим на распределение корней характеристического уравнения

$$f(\mu) = -tg(\mu) + \mu = 0;$$

построим график левой части этого уравнения:

```
>plot(f,mu=0..Pi,-0.5..0.5,discont=true,color=black);
```



Рис. 1. График функции $f(\mu)$

Найдем первые три корня характеристического уравнения:

```
>K:=3:mu:=array(1..K):
>for i from 1 to K do
>mu[i]:=fsolve(f=0,mu,mu=i*Pi..(i+1)*Pi):
>print(mu[i]);
>end do:
```

$$4.493409458$$
$$7.725251837$$
$$10.90412166$$

Пример 2. Найти собственные значения и собственные функции задачи Штурма-Лиувилля:

$$y'' + \lambda y = 0, y(a) = 0, y'(b) = 0, x \in [a, b].$$

Решение. Зададим уравнения при $\lambda \neq 0$

```
>restart:eq:=diff(y(x),x,x)+lambda*y(x)=0;
```

$$eq := \left(\frac{d^2}{dx^2} y(x)\right) + \lambda y(x) = 0$$

Находим общее решение уравнения:

```
>dsolve(eq,y(x));y:=unapply(rhs(%),x);
```

$$y(x) = \_C1 \sin(\sqrt{\lambda} x) + \_C2 \cos(\sqrt{\lambda} x)$$
$$y := x \to \_C1 \sin(\sqrt{\lambda} x) + \_C2 \cos(\sqrt{\lambda} x)$$

Задаем граничные условия:

```
>assume(b>a):
>eq1:=y(a)=0;eq2:=D[1](y)(b)=0;
```

$$eq1 := \_C1 \sin(\sqrt{\lambda} \tilde{a}) + \_C2 \cos(\sqrt{\lambda} \tilde{a}) = 0$$
$$eq2 := \_C1 \cos(\sqrt{\lambda} \tilde{b}) \sqrt{\lambda} - \_C2 \sin(\sqrt{\lambda} \tilde{b}) \sqrt{\lambda} = 0$$

Формируем матрицу коэффициентов и вычисляем ее определитель:

```
>with(linalg):genmatrix({eq1,eq2},{_C1,_C2});A:=det(%);
>Delta:=combine(%));
```

$$A := \begin{bmatrix} \sin(\sqrt{\lambda} \tilde{a}) & \cos(\sqrt{\lambda} \tilde{a}) \\ \cos(\sqrt{\lambda} \tilde{b}) \sqrt{\lambda} & -\sin(\sqrt{\lambda} \tilde{b}) \sqrt{\lambda} \end{bmatrix}$$

$$\Delta := -\sqrt{\lambda} \cos(\sqrt{\lambda} \tilde{a} - \sqrt{\lambda} \tilde{b})$$

Приравниваем нулю этот определитель и решаем полученное характеристическое уравнение:

```
>Delta:=select(has,Delta,[cos]);
```

$$\Delta := \cos(\sqrt{\lambda}\tilde{a} - \sqrt{\lambda}\tilde{b})$$

```
>_EnvAllSolutions:=true:lambda:=solve(Delta,lambda);
```

$$\lambda := \frac{\pi^2\,(1 + 2\,\_Z1\tilde{})^2}{4\,(-\tilde{b} + \tilde{a})^2}$$

```
>lambda:=subs(_Z1='k',lambda);
```

$$\lambda := \frac{\pi^2\,(1 + 2\,k)^2}{4\,(-\tilde{b} + \tilde{a})^2}$$

Находим собственные функции:

```
>assume(k,posint):y(x);
```

$$\_C1 \sin\left(\frac{\sqrt{4}\,\sqrt{\dfrac{\pi^2\,(1+2\,\tilde{k})^2}{(-\tilde{b}+\tilde{a})^2}}\,x}{4}\right) + \_C2 \cos\left(\frac{\sqrt{4}\,\sqrt{\dfrac{\pi^2\,(1+2\,\tilde{k})^2}{(-\tilde{b}+\tilde{a})^2}}\,x}{4}\right)$$

```
>C1:=solve(eq1,_C1);
```

$$C1 := \frac{\_C2\,cos(\frac{\pi\,(1+2\,\tilde{k})\,\tilde{a}}{2\,(-\tilde{b}+\tilde{a})})}{sin(\frac{\pi\,(1+2\,\tilde{k})\,\tilde{a}}{2\,(-\tilde{b}+\tilde{a})})}$$

```
>simplify(subs(_C1=C1,y(x))):combine(%);
```

$$\frac{\_C2\,\sin(\frac{\pi\,\tilde{a} - \pi\,x + 2\,\pi\,\tilde{a}\,\tilde{k} - 2\,\pi\,x\,\tilde{k}}{-2\,\tilde{b} + 2\,\tilde{a}})}{\sin(\frac{\pi\,\tilde{a} + 2\,\pi\,\tilde{a}\,\tilde{k}}{-2\,\tilde{b} + 2\,\tilde{a}})}$$

```
>Yn:=unapply(select(has,%,[x]),x,k);
```

$$Yn := (x,\,\tilde{k}) \to \sin(\frac{\pi\,\tilde{a} - \pi\,x + 2\,\pi\,\tilde{a}\,\tilde{k} - 2\,\pi\,x\,\tilde{k}}{-2\,\tilde{b} + 2\,\tilde{a}})$$

Проверим дифференциальное уравнение:

```
>y:='y':Yn(x,k):simplify(subs(y(x)=%,eq));
```

$$0 = 0$$

Проверим граничные условия:

```
>Yn(a,k)=0;simplify(D[1](Yn)(b,k))=0;
```

$$0{=}0$$
$$0{=}0$$

Проверим ортогональность собственных функций на отрезке [a,b]:

```
>assume(n,posint):assume(m,posint):
>Int(Yn(x,n)*Yn(x,m),x=a..b);simplify(value(%));
```

$$\int_{\tilde{a}}^{\tilde{b}} \sin(\frac{\pi\,\tilde{a} - \pi\,x + 2\,\pi\,\tilde{a}\,\tilde{n} - 2\,\pi\,x\,\tilde{n}}{-2\,\tilde{b} + 2\,\tilde{a}})\,\sin(\frac{\pi\,\tilde{a} - \pi\,x + 2\,\pi\,\tilde{a}\,\tilde{m} - 2\,\pi\,x\,\tilde{m}}{-2\,\tilde{b} + 2\,\tilde{a}})\,dx$$

$$0$$

Вычислим норму собственных функций:

```
>Norma:=Int(Yn(x,n)^2,x=a..b):simplify(value(%));
```

$$\frac{\tilde{b}}{2} - \frac{\tilde{a}}{2}$$

Можно преобразовать аргумент у собственных функций к более удобному виду:

```
>simplify(collect(
>(Pi*a-Pi*x+2*Pi*a*k-2*Pi*x*k)/(-2*b+2*a),x));
```

$$-\frac{(1 + 2\,\tilde{k})\,\pi\,(\tilde{a} - x)}{2\,(-\tilde{b} + \tilde{a})}$$

Таким образом, собственные значения задачи будут

$$\lambda_k = \frac{(2k+1)^2\pi^2}{4(b-a)^2},\ k = 1, 2, 3, \ldots,$$

а собственные функции —

$$y_k(x) = \sin(\frac{(2k+1)\pi(a-x)}{2(b-a)}),\ k = 1, 2, 3, \ldots.$$

Рассмотрим случай $\lambda = 0$:

```
> lambda:=0;eq;
```

$$\lambda := 0$$

$$\frac{d^2}{dx^2}\,\mathrm{y}(x) = 0$$

```
>dsolve(eq,y(x));assign(%):y0:=unapply(y(x),x);
```

$$\mathrm{y}(x) = \_C1\,x + \_C2$$

$$y0 := x \to \_C1\,x + \_C2$$

```
>eq0_1:=y0(a)=0;eq0_2:=D(y0)(b)=0;
```

$$eq0\_1 := \_C1\,a\tilde{} + \_C2 = 0$$

$$eq0\_2 := \_C1 = 0$$

```
>genmatrix({eq0_1,eq0_2},{_C1,_C2});
```

$$\begin{bmatrix} 1 & 0 \\ a\tilde{} & 1 \end{bmatrix}$$

```
>det(%);
```

$$1$$

Так как определитель отличен от нуля, значит, существует только тривиальное решение. Следовательно, $\lambda = 0$ – не есть собственное значение задачи.

Учебное издание

Лилия Александровна Молчанова

ЗАДАЧА ШТУРМА-ЛИУВИЛЛЯ

Методические указания для студентов
математических специальностей

В авторской редакции
Технический редактор Л.М. Гурова
Компьютерный набор и верстка автора

Подписано в печать 31.01.06
Формат 60 × 84 1/16. Усл. печ. л. 0.9. Уч.-изд. л. 0,8.
Тираж 25 экз.

Издательство Дальневосточного университета
690950, Владивосток, ул. Октябрьская, 27.
Отпечатано в лаборатории
кафедры компьютерных наук ИМКН ДВГУ
690950, Владивосток, ул. Октябрьская, 27, к. 132.